Figure 1: Chebyshevbpf –  $N^{\text{th}}$  Order Chebyshev Bandpass Filter

Form: Chebyshevbpf:<instance name> n1 n2 n3 <parameter list>

n1 is the input terminal

n2 is the reference terminal

n3 is the output terminal

Parameters:

Parameter	Type	Default Value	Required?
n: Filter order (must be odd)	INTEGER	11	yes
f0: Center Frequency of the filter (Hz)	DOUBLE	N/A	yes
bw: Bandwidth of the filter (Hz)	DOUBLE	N/A	yes
z0: Input/Output Impedance (ohms)	DOUBLE	50	yes
ripple: Passband ripple (dB)	DOUBLE	0.1	yes
q: resonator Q	DOUBLE	10 000	no

Example:

Chebyshevbpf:c1 2 3 0 n=5 f0=1e6 bw=400e3 z0=50 ripple=1

Notes:

The filter has a maximum allowable order of 100.

Details:

This model designs an odd-order Type-2 Cauer topology chebyshev bandpass filter, using parallel inductors and capacitors and series inductors and capacitors in a ladder structure. The filter can be of any order up to a maximum order of 100. The filter has a center frequency, f0, a bandwidth, bw, and a passband ripple, ripple. The filter is designed so that the input and output impedance, z0, of the filter is the same.

*Model Description:*

The model actually works by creating individual “inductor” and “capacitor” elements and then connecting them together in the proper ladder structure to obtain the desired filter parameters. The model begins by calculating the coefficients of an  $n$ -order chebyshev lowpass prototype filter normalized to a radian corner frequency of 1 radian/sec and a  $1\Omega$  system impedance. This is accomplished by using Equations 1-6 below.

$$\begin{aligned}
 g_0 &= 1 \\
 g_{n+1} &= \begin{cases} 1, & n \text{ odd} \\ \tanh^2(\beta/4), & n \text{ even} \end{cases} \\
 g_1 &= \frac{2a_1}{\gamma} \\
 g_k &= \frac{4a_{k-1}a_k}{b_{k-1}g_{k-1}}, \quad k = 2, 3, \dots, n \\
 a_k &= \sin \left[ \frac{(2k-1)\pi}{2n} \right], \quad k = 1, 2, \dots, n \\
 \gamma &= \sinh \left( \frac{\beta}{2n} \right) \\
 b_k &= \gamma^2 + \sin^2 \left( \frac{k\pi}{n} \right) \quad k = 1, 2, \dots, n \\
 \beta &= \ln \left[ \coth \left( \frac{R_{dB}}{17.37} \right) \right] \\
 R_{dB} &= 10 \log (\epsilon^2 + 1)
 \end{aligned}$$

The model then takes these coefficients and uses them to calculate the Inductor and Capacitor component values to build the type-2 Cauer ladder topology structure shown in Figure 1. These capacitor and inductor values are obtained from the chebyshev lowpass prototype coefficients by using impedance transformation, frequency transformation, and lowpass-to-bandpass filter transformation methods.

The L and C values in Figure 1 are calculated using the equations shown below. In Eq. 7 and 8:  $\omega_0$  is the center frequency of the filter in rad/sec;  $\omega_{BW}$  is the bandwidth of the filter in rad/sec; and  $Z_0$  is the input/output impedance of the filter.

$$C_r = \begin{cases} \frac{g_r}{\omega_{BW} Z_0} & r = \text{odd} \\ \frac{\omega_{BW}}{\omega_0^2 g_r Z_0} & r = \text{even} \end{cases} \quad L_r = \begin{cases} \frac{\omega_{BW} Z_0}{\omega_0^2 g_r} & r = \text{odd} \\ \frac{g_r Z_0}{\omega_{BW}} & r = \text{even} \end{cases}$$

Once the inductor and capacitor component values are calculated the model then begins building up the filter by actually creating each element individually and connecting them together one-by-one. The inductors are created using the “inductor” element and the capacitors are created using the “capacitor”

element. The odd numbered elements are connected in parallel, and the even numbered elements are connected in series, as shown in Figure 1. Finite resonator Q is modeled using a resistor in series with a series LC resonator and in shunt with a shunt LC resonator.

Once all of the elements have been added and connected together the model is complete. This linear model is used for both the time domain and frequency domain

#### Sample Netlists:

Example of an AC Analysis of a 5th-order filter with a center frequency of 1MHz, a bandwidth of 400kHz, a passband ripple of 1dB, and an input/output impedance of 50Ω.

**Netlist File:** chebyshevbpfactest.net

```
**** AC chebyshevbpf test ****

.ac start=10e3 stop=100e6 n_freqs=4000

vsource:vin 1 0 vac=1.0



R:Rin1 1 2 r=50
Chebyshevbpf:bl 2 3 0 n=5 f0=1e6 bw=400e3 z0=50 ripple=1
R:Rout1 3 0 r=50


.options gnuplot
.options preamble1="set logscale x; set term x11 font 'helvetica,13';
set title 'Chebyshev Filter Magnitude Response'; set xlabel 'FREQUENCY (MHz)';
set ylabel 'MAGNITUDE (dB)'"
.out plot term 3 vf term 1 vf div mag db 1e-6 scalex preamble1 in "chebyshevbpf.ac.mag"

.options preamble2="set logscale x; set term x11 font 'helvetica,13';
set title 'Chebyshev Filter Phase Response'; set xlabel 'FREQUENCY (MHz)';
set ylabel 'phase (DEGREES)'"
.out plot term 3 vf term 1 vf div prinphase 1e-6 scalex rad2deg preamble2 in
"chebyshevbpf.ac.phase"

.end

***** FREEDA 1.3 stopping on Thu Apr 17 19:29:09 2008 *****
```





Example of a Transient Analysis (.TRAN2) of a 5th-order filter with a center frequency of 1MHz, a bandwidth of 400kHz, a passband ripple of 1dB, and an input/output impedance of  $50\Omega$ . The source frequency is set to the center frequency of 1MHz, with a 1V pk-to-pk amplitude.

**Netlist File:** chebyshevbpfrantest1.net

```
**** tran2 chebyshevbpf test #1 ****


.tran2 tstop=40us timestep=1ns out_steps=40000

vsource:vin 1 0 vac=1.0 f=1e6

R:Rin1 1 2 r=50
Chebyshevbpf:b1 2 3 0 n=5 f0=1e6 bw=400e3 z0=50 ripple=1
R:Rout1 3 0 r=50

.options gnuplot
.options preamble1="set term x11 font 'helvetica,13';
set title 'Chebyshev Filter Transient Response'; set xlabel 'TIME (microseconds)';
set ylabel 'Voltage (V)'
.out plot term 3 vt 1e6 scalex preamble1 in "chebyshevbpf.tran.out1"

.end
```



Example of a Transient Analysis (.TRAN2) of a 5th-order filter with a center frequency of 1MHz, a bandwidth of 400kHz, a passband ripple of 1dB, and an input/output impedance of 50 $\Omega$ . The source frequency is set outside of the bandwidth of the filter at 1.5MHz, with a 1V pk-to-pk amplitude.

**Netlist File:** chebyshevpftrantest2.net

```
**** tran2 chebyshevpf test #2 ****


.tran2 tstop=40us tstep=1ns out_steps=4000

vsource:vin 1 0 vac=1.0 f=1.5e6

R:Rin1 1 2 r=50
Chebyshevpf:b1 2 3 0 n=5 f0=1e6 bw=400e3 z0=50 ripple=1
R:Rout1 3 0 r=50

.options gnuplot
.options preamble1="set term x11 font 'helvetica,13';
set title 'Chebyshev Filter Transient Response'; set xlabel 'TIME (microseconds)';
set ylabel 'Voltage (V)'"
.out plot term 3 vt 1e6 scalex preamble1 in "chebyshevpf.tran.out2"

.end
```




---

#### References:

M. Steer and W. Fathelbab, "Filters" in *Microwave and RF Design*.

---

#### Version:

2008.05.11 (2008 May 11)

---

#### Credits:

Name:

Shawn Evans

[sdevans@ncsu.edu](mailto:sdevans@ncsu.edu)

Michael Steer

[mbs@ncsu.edu](mailto:mbs@ncsu.edu)

Affiliation

NC State University

NC State University

Date

April 2008

May 2008

Links

**NC STATE UNIVERSITY**

[www.ncsu.edu](http://www.ncsu.edu)

**NC STATE UNIVERSITY**

[www.ncsu.edu](http://www.ncsu.edu)